
Wer kennt nicht solche Spieleklassiker wie 
Raiden oder Raptor: Call of the Shadows? 
Der Spieler steuert ein Raumschiff, wäh-
rend von oben eine Gegnerwelle nach 
der anderen auf ihn zukommt. Solch ein 
Ballerspiel zu programmieren ist nicht 
schwer: Wie bei jeder Programmieraufga-
be muss das große Problem in viele kleine 
Probleme zerlegt werden, die der Reihe 
nach gelöst werden. In diesem Artikel 
wird gezeigt, wie Sie die Steuerung für das 
Heldenraumschiff programmieren, Geg-
ner zufällig hereinkommen lassen, wie Sie 
Ihrem Raumschiff zu Munition verhelfen 
und wie Sie die Kollisionsabfrage beim 
Zusammenstoß mit Gegnern nutzen.

Laden Sie sich zunächst einmal die Datei 
shooter.­zip von der CD herunter (oder von 
www.freimark.de/training/mx05-shooter 
.zip) und entpacken Sie diese. Sie finden dar-
in alle notwendigen Grafiken und die offene 
FLA-Datei in verschiedenen Stadien, von 
shooter-01start.fla über shooter-03.fla bis 
zu shooter-05final.fla. Im Text wird dann 
an wichtigen Stellen erwähnt, welches Sta-
dium wir erreicht haben. Dadurch können 
Sie schnell wieder Anschluss finden oder in 
einer späteren Version etwas nachsehen, 
falls etwas unklar ist.

Hey ho, let’s go! Öffnen Sie die Da-
tei shooter-01start.fla. Die Datei enthält 
eine leere Zeitleiste, allerdings sind be-
reits alle notwendigen Grafiken in der 
Bibliothek vorhanden. Zunächst sollten 
Sie von vornherein für Ordnung in Ihrem 
Flash-Projekt sorgen: Legen Sie sich da-
her die folgenden fünf Ebenen an: Hinter-
grund, Items, Messages, Sound, Actions 
und Label. Klicken Sie in der Ebene Label 
in das erste Bild und geben Sie start als 
Bildmarkierung in der Eigenschaften-
palette ein. Wählen Sie die Hintergrund
ebene an und ziehen Sie die Bitmap bg.gif 
von der Bibliothek auf die Bühne. Die 
Koordinaten für x und y sind jeweils 0 
(von der linken oberen Ecke der Instanz 
aus). Gehen Sie zu Bild 10, markieren Sie 
mit gedrückter Shift-Taste die Bilder 
aller Ebenen und wählen Sie den Befehl 
Modifizieren | Zeitleiste | In Schlüs-
selbilder konvertieren oder drücken 
Sie die Taste [F6] (etwas Abstand zur vor-
herigen Bildmarkierung sorgt für Über-
sicht). Als Bildmarkierung geben Sie bitte 
game ein. Wiederholen Sie diese Schritte 
im Abstand von zehn Bildern für die Bild-
markierungen level1 und game_over, bis 
Ihre Datei so aussieht wie in Abbildung 1. 

Nicht vergessen: Erfolgserlebnisse früh-
zeitig und häufig abspeichern!

Bevor das eigentliche Spiel bei der 
Bildmarkierung level1 beginnt, sollten 
Sie dem Spieler bei der Markierung game 
eine kurze Anleitung an die Hand geben 
sowie eine Möglichkeit, das Spiel selbst zu 
starten, damit er sich nicht unvermittelt im 
Spielgeschehen wiederfindet. Hierzu ge-
nügt in der Ebene Messages ein einfacher 
Button mit einer Beschriftung wie Go!, der 
zur Bildmarkierung level1 springt, sobald 
er gedrückt wird. Es ist guter Stil, sämt-
lichen Code in Frame-Skripts abzulegen, 
statt ihn direkt auf Buttons und Movie-
Clips zu schreiben. Öffnen Sie daher an 
der Bildmarkierung game in der Ebene 
Actions die Aktionenpalette (Fenster | 
Entwicklungsbedienfelder | Akti-
onen oder die Taste [F9]). Fügen Sie dann 
folgenden Code ein:

stop();

go_button.onRelease = function () {

   gotoAndPlay ("level1");

};

Die stop()-Action ist notwendig, damit der 
Film an dieser Stelle nicht einfach weiter-

läuft. Denken Sie bitte daran, dass der But-
ton erst funktioniert, wenn Sie ihm in der 
Eigenschaftenpalette einen Instanznamen 
gegeben haben (in diesem Fall go_button). 
Instanznamen sind notwendig, um Symbole 
per ActionScript ansprechen zu können.

Raumschiffkapitän. So weit die Vor-
bereitungen. Nun kümmern wir uns um 
die Steuerung des Raumschiffs. Wählen 
Sie die Items-Ebene und gehen Sie zur 
Bildmarkierung level1. Ziehen Sie die 
Bitmap ship.gif von der Bibliothek auf die 
Bühne; die Position spielt keine Rolle. Da 
nur MovieClip-Symbole in Flash animiert 
werden können, müssen Sie die Bitmap 
mit Modifizieren | In Symbol konver-
tieren ([F8]) in ein solches verwandeln. 
Vergeben Sie einen aussagekräftigen Na-
men wie zum Beispiel ship_mc (generell 
sollten Sie Unterstriche statt Bindestriche 
verwenden, da Flash den Bindestrich als 
Subtraktionsoperator interpretiert). Die 
Erweiterung _mc wurde verwendet, da 
das Syntax-Highlighting von Flash dann 
erkennt, dass es sich um einen MovieClip 
handelt. Es ist zu empfehlen, Ordnung in 
der Bibliothek zu halten, indem Sie einen 
Ordner MCs anlegen und die MovieClip-
Symbole dorthin verschieben, statt sie lose 
in der Bibliothek zu belassen. Wenn Sie in 
größeren Projekten mit 200 oder mehr 
Bitmaps und Symbolen in der Bibliothek 
zu tun haben oder eine fremde Datei in-
spizieren müssen, werden Sie das Ordner-
Feature schätzen lernen.

Ergänzen Sie auch bei Ihrem Raum-
schiff den Instanznamen (vergeben Sie 

einfach den gleichen Namen, den Sie 
bei der Umwandlung der Bitmap in ei-
nen MovieClip verwendet haben, also 
ship_mc). Klicken Sie dann in das Bild 
der Actions-Ebene und öffnen Sie die 
Aktionenpalette, um folgenden Code 
einzugeben:

Mouse.hide();

stop ();

ship_mc.onEnterFrame = function () {

   this._x = _root._xmouse - 36;

   this._y = _root._ymouse - 45;

} ;

Mit Mouse.hide() schalten Sie den 
Mauscursor aus, der sonst über dem 
Raumschiff etwas stören würde. Die 
stop()-Action sorgt auch hier dafür, dass 
der Abspielkopf der Hauptzeitleiste nicht 
einfach weiterläuft. Im Raumschiff-Mo-
vieClip werden wegen des onEnterFrame 
dennoch permanent die x- und die y-Ko-
ordinaten des MovieClips an die aktuellen 
Koordinaten der Maus angepasst – ein 
klassischer Mausverfolger. Da die Koordi-
naten hier von der linken oberen Ecke des 
MovieClips aus gemessen werden, ist noch 
eine Korrektur um 36 beziehungsweise 45 
Pixel eingebaut, sodass der Mausverfolger 
tatsächlich für die Mitte des Raumschiffs 
gilt. Wichtig: Vor den Eigenschaften _x 
und _y müssen Sie unbedingt noch this 
notieren, da Sie sonst die Koordinaten der 
Bühne verschieben (denn die Hauptzeit-
leiste _root ist auch ein MovieClip, und 
dieses Skript steht in einem Frame-Skript 
von _root).

Kanonenfutter. So allein auf dem Spiel-
feld kommt noch keine Spannung auf. Da-
her werden wir nun für die Gegnerwellen 
sorgen. Falls es zu Problemen gekommen 
sein sollte, finden Sie jetzt mit der Datei 
shooter-02.fla wieder Anschluss. Ziehen 
Sie enemy.gif von der Bibliothek auf die 
Bühne (in das gleiche Bild und die gleiche 
Ebene wie Ihr Raumschiff). Die Position 
spielt keine Rolle. Wandeln Sie das Geg-
nerschiff in ein MovieClip-Symbol um 
und tragen Sie auch gleich den Namen des 
MovieClips als Instanznamen ein, bei-
spielsweise enemy_mc. Dieser MovieClip 
ist die Vorlage für alle Gegnerraumschiffe, 
sodass wir nicht für Dutzende Gegner ein-
zelne Bitmaps ablegen müssen. Der Vorla-
ge-MovieClip selbst soll nicht angezeigt 
werden. Notieren Sie daher 

enemy_mc._visible = false;

Schreiben Sie innerhalb des enemy_mc-
MovieClips, wo die Kopien der Vorlage 
erscheinen sollen (klicken Sie dazu dop-
pelt auf den MovieClip, um zu dessen 
Zeitleiste zu gelangen):

this._y = -100;

this._x = random(Stage.width);

stop() ;

Die Gegner erscheinen vertikal knapp 
über der Bühne und horizontal zufällig 
verteilt. Das stop(); bewirkt auch hier ein 
Anhalten der Zeitleiste (jetzt gibt es zwar 
nur ein Bild, aber später brauchen wir das 
noch). In Abbildung 2 sehen Sie, wie Sie 
vorgehen müssen.

Klicken Sie nach der Skript-Eingabe 
wieder auf Szene 1, um zur Hauptzeit-
leiste zurückzukehren. Jetzt ist die Vor-
lage wie gewünscht unsichtbar, aber 
Gegner brauchen wir immer noch. Hier 
der Code des Hauptskripts, das sich bei 
der Bildmarkierung level1 in der Ebene 
Actions befindet (im Folgenden nur noch 
„Hauptskript“). Bitte fügen Sie die fett 
geschriebenen Zeilen aus Listing 1 Ihrem 
Skript hinzu.

Die erste der drei neuen Zeilen zählt 
einen Timer hoch. Die zweite Zeile ruft 
eine Funktion mit dem Namen initialize
Items auf und übergibt ihr zwei Para-
meter, in diesem Fall enemy und 40. Der 
Code der Funktion folgt gleich. Zuerst 
soll erklärt werden, weswegen wir die 
beiden Zeilen an dieser Stelle eingefügt 
haben: Stünden sie außerhalb der Funk-

Abbildung 1: So 
sieht die Datei zu 
Anfang aus.

tion, auf einer Ebene wie Mouse.hide ();, 
würden die Zeilen nur einmal ausgeführt, 
da der Abspielkopf nur einmal das Skript 
liest. Für das Hochzählen des Timers ist 
es notwendig, dass die Zeile fortlaufend 
ausgeführt wird (wir kommen gleich 
darauf, wofür der Zähler überhaupt gut 
ist). Da uns das Raumschiff immer zur 
Verfügung steht und wegen des Mausver-
folgers sowieso schon eine EnterFrame-
Funktion hat, können wir den Code gut 
an dieser Stelle unterbringen. Die zweite 
Zeile mit dem initializeItems-Funkti-
onsaufruf muss zwar nicht permanent 
ausgeführt werden, aber doch mehr als 
einmal. Daher steht sie ebenfalls inner-
halb des EnterFrame. Die dritte Zeile 
setzt den Zähler auf Null. Das soll nur 
ein einziges Mal gemacht werden, und 
deswegen steht dieser Code außerhalb 
des EnterFrame. Nun zur Funktion, die 
Sie bitte direkt unterhalb des bisherigen 
Codes notieren (Listing 2).

Zugegeben: Sieht ein bisschen wild 
aus. Was passiert hier? Die Funktion 
nimmt zwei Parameter entgegen, item und 
intervall_length. Der erste Parameter teilt 
der Funktion mit, um welches Symbol es 

geht. Im Moment geht es nur um die geg-
nerischen Schiffe, aber später können wir 
den gleichen Code für Goodies wie einen 
Gesundheitsbonus verwenden (wenn Sie 
Code mehr als einmal benötigen, sollten 
Sie ihn generell in eine Funktion ausla-
gern). In unserem Beispiel arbeitet die 
Funktion zunächst mit den Parametern 
enemy als item und 40 als intervall_length. 
Das Intervall wird innerhalb des Enter
Frame unseres Raumschiffs hochgezählt. 
Damit die Gegner nur etwa alle 40 Bilder 
erscheinen und nicht exakt nach 40 Bil-
dern, wird der zeitliche Abstand mit der 
Zeile intervall = random (intervall_length) 
+ intervall_length; etwas verändert. Im 
folgenden if-Statement wird geprüft, ob 
es sich um Gegnerraumschiffe handelt. 
item.indexOf gibt die Position des ersten 
Auftauchens des gesuchten Strings zu-
rück. indexOf lautet -1, wenn der String 
nicht gefunden wird; genau das, was wir 
benötigen.

Der restliche Code der Funktion steht 
in einer weiteren if-Abfrage und wird nur 
ausgeführt, wenn der Zähler für das über-
gebene Item überschritten wird. Dann 
passiert Folgendes: Der Zähler wird zu-
rückgesetzt, die Variable _root.zindex 
wird um eins hochgezählt und es wird eine 
neue Variable new_name erzeugt (die aus 
dem übergebenen Item und dem _root.
zindex besteht). In der vorletzten Zeile 
wird das Vorlagen-MovieClip dupliziert: 
Der Name der Vorlage wird dabei „dy-
namisch“ zusammengesetzt – this[item + 
"_mc"] (ohne Punkt nach this!) wird wäh-
rend des Programmablaufs ersetzt durch 

den item-Parameter, den die Funktion 
mitbekommen hat. Außerdem wird per 
String-Verknüpfung noch _mc hinzuge-
fügt. Aus this[item + "_mc"] wird also 
this.enemy_mc.

Diese dynamischen Pfade benöti-
gen wir in der letzten Zeile gleich noch 
einmal. Aber erst noch kurz zurück zur 
vorletzten Zeile: duplicateMovieClip 
ist eine MovieClip-Methode, die eine 
Instanz (eine Kopie) des angegebenen 
MovieClips im Speicher erzeugt. Als Pa-
rameter benötigt diese Methode einen 
neuen Namen, den wir hier mit new_
name einsetzen, sowie eine Tiefe, hier 
der zindex-Zähler. Auf einer Tiefenebene 
kann sich immer nur ein MovieClip be-
finden. Wenn wir die Gegnerschiffe auf 
der gleichen Ebene duplizieren würden, 
befände sich immer nur ein Gegnerschiff 
auf der Bühne. Die letzte Zeile Action
Script bedeutet: „Du MovieClip da drü-
ben, mach’ jetzt das, was in der Funktion 
enemyActions() steht, und mach’ es am 
laufenden Band!“ Der dynamische Pfad 
this["enemy" + root.zindex] spricht ein 
MovieClip auf der Bühne an, dessen 
Instanzname mit enemy beginnt, gefolgt 
von der Zahl, die im Moment des Funk-
tionsaufrufs in _root.zindex steht.

Da wir mit den Gegnerschiffen noch 
viel vorhaben, wird deren Code in eine 
eigene Funktion ausgelagert, und zwar 
enemyActions(), denn sonst wird es in 
der Funktion initializeItems() etwas 
unübersichtlich. Schreiben Sie bit-
te folgenden Code ganz unten in das 
Hauptskript:

Abbildung 2:  
Schalten Sie die Vorla-

ge für die Gegnerraum-
schiffe aus, indem Sie 
sie im ersten Bild des 

MovieClips unsichtbar 
machen.

Listing 1
Mouse.hide ();

stop ();

ship_mc.onEnterFrame = function () {

   this._x = _root._xmouse - 36;

   this._y = _root._ymouse - 45;

   _root.item_timer_enemy++;

   initializeItems ("enemy", 40);

};

enemy_mc._visible = false;

_root.item_timer_enemy = 0;

Listing 2
function initializeItems (item, intervall_length) {

  var intervall = random (intervall_length) +

  intervall_length;

    if (item.indexOf ("enemy") != -1) {

    if (this["item_timer_" + item] >=

      intervall) {

      this["item_timer_" + item] = 0;

      _root.zindex++;

      var new_name = item + _root.zindex;

      this[item + "_mc"].duplicateMovieClip

        (new_name, _root.zindex);

        this["enemy" + _root.zindex].

          onEnterFrame = enemyActions;

    }

  }

}

function enemyActions () {

  this._y += 10;

  if (this._y > Stage.height) {

    this.removeMovieClip ();

  }

}

Mit this._y += 10; bewegen Sie das Geg-
nerschiff – genauer: die Instanz des Geg-
nerschiffs – über die Bühne. Wenn der 
Gegner unten aus der Bühne fliegt, wird 
die Instanz von der Bühne gelöscht. Diese 
Zeile sollten Sie unbedingt verwenden: 
Zwar können Sie das Gegnerschiff nicht 
mehr sehen, aber es ist noch vorhanden. 
Bei einigen Dutzend Schiffen – auch au-
ßerhalb der Bühne – wirkt sich das durch-
aus auf die Performance aus.

Kollisionsalarm. So weit, so gut. Falls 
es zu Problemen gekommen ist, können 
Sie in der Datei shooter-03.fla nachsehen 
oder mit dieser Datei weiterarbeiten.

Wenn Sie den Film testen, wird Ih-
nen auffallen, dass Sie durch die gegne-
rischen Schiffe hindurchfliegen können 
und diese das völlig unbeeindruckt lässt. 
Der Grund dafür ist, dass wir noch keine 
Kollisionsabfrage eingebaut haben: Die 
Raumschiffe wissen noch gar nichts von-
einander. In diesem Abschnitt werden wir 
sie einander bekannt machen. Zunächst 
statten wir unser Spielerraumschiff mit 
Gesundheitspunkten aus und legen fest, 
wie viele Punkte beim Zusammenstoß mit 

einem Gegner abgezogen werden. Schrei-
ben Sie zu Beginn des Hauptskripts _root.
health = 100; und fügen Sie dem Skript 
innerhalb des enemy_mc die Zeile this.
damage_collision = 25; hinzu.

Ergänzen Sie nun die Funktion enemy
Actions() im Hauptskript um die fett ge-
druckten Zeilen aus Listing 3.

Zur Erklärung: Für die Kollisionsab-
frage benötigen Sie nicht mehr als die Mo-
vieClip-Methode hitTest(). Als Parameter 
geben Sie den MovieClip an, mit dem Sie 
auf Überlappung prüfen möchten. Wenn 
die Bedingung wahr ist, werden dem Spie-
ler die 25 Schadenspunkte abgezogen, die 
im ersten Bild des Gegner-MovieClips hin-
terlegt sind. Danach werden die Schaden-
spunkte auf null gesetzt, damit sie nicht 
mehrfach abgezogen werden (denn die 
Kollision dauert mehr als ein Bild an). Ab-
schließend springt der Abspielkopf inner-
halb des Gegner-MovieClips auf das Bild 
2 und spielt die Zeitleiste ab, in der das 
Gegnerschiff nach dem Treffer in Flam-
men aufgeht. Hierzu kommen wir jetzt: 
Doppelklicken Sie auf die Gegnervorlage 
enemy_mc auf der Bühne. Erstellen Sie 
in der Zeitleiste dieses MovieClips eine 
weitere Ebene und fügen Sie im zweiten 
Bild mit Modifizieren | Zeitleiste | 
In Schlüsselbilder konvertieren 
(oder der Taste [F6]) ein Schlüsselbild ein. 
Wählen Sie dann Datei | Importieren | 
In Bühne importieren (oder das Tasta-
turkürzel [Apfel + R] bzw. [Strg + R]). 
Navigieren Sie zum Ordner materials/ex-

plosion und importieren Sie das erste Bild, 
striche_f01.gif. Flash wird anhand der 
fortlaufenden Nummerierung der Bilder 
erkennen, dass es sich um eine Sequenz 
handelt und Sie fragen, ob Sie alle Bilder 
der Sequenz importieren möchten. Bestä-
tigen Sie dies. Fügen Sie der Actions-Ebe-
ne innerhalb des MovieClips bei Bild 10 
ein leeres Schlüsselbild hinzu (Einfügen 
| Zeitleiste | Leeres Schlüsselbild 
oder Taste [F7]) und geben Sie die Zeile 
this.­removeMovieClip(); auf dem Schlüs-
selbild ein. Falls beim Testen des Films die 
Explosion ständig abgespielt wird, haben 
Sie im ersten Bild des enemy_mc die stop()-
Action vergessen (siehe oben im Abschnitt 
Kanonenfutter). Alles in allem sollte es 
jetzt bei Ihnen ungefähr so aussehen wie 
in Abbildung 3.

Wenn Sie mit Ihrem Raumschiff in 
einen Gegner fliegen, werden Ihnen zwar 
Gesundheitspunkte abgezogen, aber zwei 
Dinge fehlen noch: Einmal eine Anzeige 
der verbleibenden Punkte und zum ande-
ren sollte das Spiel zu Ende sein, wenn die 
Gesundheit bei null Punkten angelangt ist. 
Für die Anzeige fügen Sie der Bühne zwei 
Textfelder auf der Ebene Messages hinzu: 
Ein statisches Textfeld, in das Sie Health 
schreiben sowie ein dynamisches Textfeld. 
In der Eigenschaftenpalette des Textfelds 
tragen Sie _root.health in das Var-Feld ein, 
damit das Textfeld mit den Inhalten der 
Variable _root.health gefüllt wird. Der 

Listing 3
function enemyActions () {

  this._y += 10;

  if (this._y > Stage.height) {

    this.removeMovieClip ();

  }

  if (this.hitTest(ship_mc)) {

    _root.health =  

      _root.health - this.damage_collision;

    this.damage_collision = 0;

    this.gotoAndPlay(2);

    }

}

Abbildung 3: 
Achten Sie darauf, 
dass sich die Ebene 
mit der Explosions-
bildfolge über der 
Ebene mit dem 
gegnerischen Schiff 
befindet.

Listing 4
if (this.hitTest (ship_mc)) {

    _root.health =  

         _root.health - this.damage_collision;

    this.damage_collision = 0;

    this.gotoAndPlay (2);

    if (_root.health <= 0) {

         gotoAndPlay ("game_over");

    }

  }

Instanzname kann frei bleiben; ihn be-
nötigen Sie nur, wenn Sie das Textfeld per 
ActionScript ansprechen möchten. Emp-
fehlung: Benutzen Sie für unser Spiel die 
Schrift „Courier New“ in Größe 22 und 
fett. Für die Ausgabe des Gesundheitssta-
tus dürfen Sie nicht vergessen, die Glyphen 
(Schriftkonturen) der Ziffern einzubin-
den (Dialogfenster Zeichenoptionen, 
das Sie über die Eigenschaftenpalette 
erreichen; siehe Abbildung 4). Wenn die 
Variable Zeichen beinhaltet, für die keine 
Glyphen-Information vorliegt, verwendet 
Flash eine generische Systemschriftart.

Um zu verhindern, dass Sie trotz null 
Gesundheit weiterfliegen können, ergän-
zen Sie bitte im Hauptskript die Funktion 
enemyActions() durch eine if-Abfrage in-
nerhalb des hitTest (Listing 4).

Wenn Sie zu oft andere Raumfahrzeuge 
rammen, wird das Spiel beendet, indem 
der Abspielkopf zur Bildmarke game_over 
springt. Dort sollten Sie mit einer entspre-
chenden Nachricht auf das Spielende hin-
weisen und eine Schaltfläche anbieten, mit 
der ein neues Spiel gestartet werden kann. 
In der Actions-Ebene bei der game_over-
Bildmarke steht dieses kurze Skript:

Mouse.show ();

stop ();

over_button.onRelease = function () {

  gotoAndPlay ("game");

};

Um dem Spieler das Anklicken des Buttons 
zu erleichtern, blenden wir in der ersten 

Zeile den Mauscursor wieder ein. Außer-
dem wird die Hauptzeitleiste angehalten, 
damit der Film nicht weiterläuft. Daran 
schließt sich der Code für den Button an, 
bei dessen Anklicken der Abspielkopf wie-
der an den Beginn des Spiels springt. 

Feuer frei! Es wird Zeit, dass Sie Ihrem 
Raumschiff ein aktives Verteidigungssys-
tem gönnen. Wenn Sie möchten, können 
Sie mit der Datei shooter-04.fla weiterar-
beiten. Bitte begeben Sie sich wieder zur 
Bildmarkierung level1.

Ziehen Sie zunächst die Bitmap bullet_
standard.gif von der Bibliothek an eine 
beliebige Position auf der Items-Ebene der 
Bühne und wandeln Sie sie in ein Movie-
Clip-Symbol um (denken Sie auch an den 
Instanznamen). Fügen Sie dem Hauptskript 
den Code aus Listing 5 hinzu.

Was macht dieser Code? In der ersten 
Zeile wurde für die Tiefenebene eine hohe 

Abbildung 4: Bei 
dynamischen Text-
feldern müssen Sie 
angeben, welche 

Zeichenformen 
Flash in die SWF-
Datei exportieren 

soll.

Listing 5
_root.bullet_zindex = 5000;

bullet_standard_mc._visible = false;

ship_mc.onMouseDown = function () {

  _root.bullet_zindex++;

  new_bullet_name = "bullet" + _root.bullet_zindex;

  bullet_mc = bullet_standard_mc.duplicateMovieClip

    (new_bullet_name, _root.bullet_zindex);

  bullet_mc._x = this._x + 36;

  bullet_mc._y = this._y;

  bullet_mc.onEnterFrame = bulletActions;

};

Zahl gewählt, damit sich die Ebenen der 
Schüsse nicht mit denen der Gegnerschiffe 
überschneiden. In Zeile zwei wird die Vor-
lage ausgeblendet, damit der ursprüngliche 
Schuss nicht auf der Bühne herumliegt. 
Daran schließt sich ein Codeschnipsel an, 
der bei einem Mausklick ausgeführt wird. 
Als Erstes wird darin der Zähler für die 
Schüsse hochgezählt, damit uns eine neue 
Tiefenebene zur Verfügung steht. Jeder 
MovieClip auf der Bühne, den wir einzeln 
per ActionScript ansprechen möchten, 
benötigt einen eigenen Instanznamen. 
Die Instanznamen der duplizierten Movie
Clips beginnen in diesem Fall mit bullet, 
gefolgt von der fortlaufenden Nummer 
aus _root.bullet_zindex. Wir hätten auch 
wie gehabt den MovieClip duplizieren 
können. Es gibt jedoch einen großen Vor-
teil, den diese Vorgehensweise im Gegen-
satz zur ersten Methode hat, die bei den 
Gegnerschiffen zum Einsatz kam: Wir 
müssen uns nicht um die relative Pfadan-
gabe kümmern. Falls Sie es dennoch wie 
vorhin probieren möchten, müssen Sie 
this._parent verwenden; im Hauptskript 
der Datei shooter-05final.fla ist in einem 
Kommentar die alternative Vorgehens-
weise hinterlegt.

Die beiden folgenden Zeilen positio-
nieren den Schuss, sodass er an der Spitze 
des Raumschiffs erscheint. Anschließend 
wird onEnterFrame der Inhalt der Funk-
tion bulletActions ausgeführt, den Sie in 
Listing 6 sehen.

Aus Gründen der Übersicht wurde 
dieser Code aus der onMouseDown-
Funktion herausgehalten. Der Schuss 

wird mit einer Geschwindigkeit von 20 
Pixeln pro Bild über die Bühne bewegt 
und beim Verlassen der Bühne gelöscht. 
In der for-Schleife findet die eigentliche 
Action statt: In einer ersten if-Abfrage 
wird geprüft, ob es sich beim MovieClip, 
den sich die Schleife gerade ansieht, um 
einen Gegner handelt. Falls ja, wird eine 
Kollisionsabfrage ausgeführt, die an die 
zusätzliche Bedingung geknüpft ist, dass 
der Gegner nicht gerade am Explodieren 
ist (ein Treffer soll schließlich nur einmal 
gezählt werden). 

Die Trefferpunkte werden hinzu ad-
diert, die Schadpunkte im Gegner auf null 
gesetzt, die Explosion im Gegnerschiff 
wird abgespielt und der Schuss wird 
entfernt. In das Skript bei der Bildmarke 
game sollten Sie noch ein _root.points = 0; 
einfügen, damit jedes neue Spiel mit null 
Punkten auf dem Konto beginnt.

Fügen Sie an der Bildmarke level1 in 
der Ebene Messages noch ein statisches 
und dynamisches Textfeld für die Ausgabe 
von _root.points ein, so wie Sie es vorhin 
bereits mit der Anzeige der Gesundheit 
gemacht haben.

Schicht im Schacht. Unser Spiel funk-
tioniert schon ganz gut. Allerdings läuft 
es unendlich, sofern der Spieler nicht 
allzu oft gegen feindliche Raumschiffe 
stößt. Die einfachste Möglichkeit, für ein 
Spielende oder den Sprung in den nächs-
ten Level zu sorgen, ist wieder ein Zähler. 
Fügen Sie die Zeile _root.zeit = 59 * 18; zu 
Beginn des Hauptskripts ein (59 mal 18 
Bilder pro Sekunde ergeben 59 Sekunden) 
und schreiben Sie _root.zeit--; innerhalb 
des ship_mc.onEnterFrame. Bei jedem 
Durchlauf wird der Timer um eins redu-
ziert. Um das Erreichen eines bestimmten 
Werts zu prüfen, eignet sich ebenfalls ein 
EnterFrame wie dieser. Ergänzen Sie da-
her:

if (_root.zeit <= 0) {

  gotoAndPlay ("game_over");

}

_root.zeitanzeige = Math.round (_root.zeit / 18);

Die letzte Zeile können Sie verwenden, 
um den Countdown in einem weiteren 
dynamischen Textfeld auszugeben. Um 
die Sache abzurunden, sollten Sie bei der 
Bildmarke game_over nochmals die An-
zeige der Punkte einfügen (kopieren Sie 
einfach das statische und das dynamische 
Textfeld für die Punktanzeige von der 

Bildmarke level1 und fügen Sie diese bei 
game_over ein).

Unser Spiel ist jetzt fertig, zumindest 
für diesen Workshop. Abschließend noch 
einige Vorschläge, wie Sie das Projekt 
selbst erweitern können: Mit der Funk-
tion initializeItems() haben Sie die Mög-
lichkeit, in späteren Levels sehr einfach 
weitere Gegner oder auch Goodies ins 
Spiel zu bringen. Wie wäre es mit einem 
Gesundheitsbonus oder Doppelfeuer für 
Gegner, die mehr als einen Treffer ein-
stecken können? Die Bilder health.gif 
und double.gif haben Sie bereits in der 
Bibliothek. Sie könnten auch ein Schutz-
schild-Goodie einführen, sodass zunächst 
die Schildpunkte aufgezehrt werden, ehe 
es an die Gesundheit geht. Sehen Sie sich 
in der Flash-Hilfe auch einmal das Kapi-
tel zum Sound an, denn im Moment ist 
das Spiel noch ein Stummfilm und erst 
mit der passenden Hintergrundmusik 
sowie Schuss- und Explosionsgeräuschen 
kommt Stimmung auf. Viel Spaß!

Listing 6
function bulletActions () {

  this._y -= 20;

if (this._y <= 0) {

    this.removeMovieClip ();

  }

  for (Objects in _root) {

    if ((Objects.indexOf ("enemy") != -1)) {

         if (this.hitTest (this._parent[Objects]) &&

           (this._parent[Objects])._currentframe == 1) {

           _root.points += 100;

           this._parent[Objects].damage_collision = 0;

           this._parent[Objects].gotoAndPlay (2);

           this.removeMovieClip ();

        }

    }

  }

}

More Info

Zur Lektüre empfehle ich Ihnen ein PDF 
von Macromedia:  
„ActionScript Coding Standards“, das 
Sie unter www.macromedia.com/ 
devnet/mx/flash/whitepapers/ 
actionscript_standards.pdf finden.

Buchempfehlungen: „Macromedia 
Flash MX 2004 Game Design De-
mystified“ von Jobe Makar und Ben 
Winiarczyk (ISBN: 0735713987), 
„ActionScript für Flash MX – Das 
Handbuch“ von Colin Moock (ISBN: 
3897213540) sowie „Sounddesign 
für das Web“ von MX-Magazin-Autor 
Holger Eggert (ISBN: 3827263468).

Stefan Freimark ist freier 
Multimedia-Producer aus 
Erlangen und hat zwei 
Laster: gute Filme und 
die griechische Küche. 
Hauptsächlich setzt er 
grafische Konzepte in 
CSS-Layouts und Flash-Websites um. 
Wenn dann noch Zeit bleibt, schreibt 
er an seinem Weblog my-two-cents.de 
weiter.

stefan@freimark.de 
www.freimark.de

